

TITLE 'ASM SYMBOL TABLE MODULE'
 SYMBOL TABLE MANIPULATION MODULE

```

1340      ORG      1340H
0200 =    IOMOD   EQU      200H      ;IO MODULE ENTRY POINT
0212 =    PCON    EQU      IOMOD+12H
021E =    EOR     EQU      IOMOD+1EH

```

ENTRY POINTS TO SYMBOL TABLE MODULE

```

1340 C3A015    JMP      ENDMOD
1343 C35C14    JMP      INISY
1346 C39E14    JMP      LOOKUP
1349 C39814    JMP      FOUND
134C C3EB14    JMP      ENTER
134F C36015    JMP      SETTY
1352 C37215    JMP      GETTY
1355 C38D15    JMP      SETVAL
1358 C39615    JMP      GETVAL

```

COMMON EQUATES

```

0078 =    PBMAX   EQU      120      ;MAX PRINT SIZE
010C =    PBUFF   EQU      10CH     ;PRINT BUFFER
0184 =    PBP     EQU      PBUFF+PBMAX ;PRINT BUFFER POINTER

0185 =    TOKEN   EQU      PBP+1    ;CURRENT TOKEN UNDER SCAN
0186 =    VALUE   EQU      TOKEN+1  ;VALUE OF NUMBER IN BINARY
0188 =    ACCLEN  EQU      VALUE+2  ;ACCUMULATOR LENGTH
0040 =    ACMAX   EQU      64       ;MAX ACCUMULATOR LENGTH
0189 =    ACCUM   EQU      ACCLEN+1

01C9 =    EVALUE  EQU      ACCUM+ACMAX ;VALUE FROM EXPRESSION ANALYSIS

01CB =    SYTOP   EQU      EVALUE+2  ;CURRENT SYMBOL TOP
01CD =    SYMAX   EQU      SYTOP+2   ;MAX ADDRESS+1

01CF =    PASS    EQU      SYMAX+2  ;CURRENT PASS NUMBER
01D0 =    FPC     EQU      PASS+1   ;FILL ADDRESS FOR NEXT HEX BYTE
01D2 =    ASPC    EQU      FPC+2    ;ASSEMBLER'S PSEUDO PC
01D4 =    SYBAS   EQU      ASPC+2   ;BASE OF SYMBOL TABLE
01D6 =    SYADR   EQU      SYBAS+2  ;CURRENT SYMBOL BEING ACCESSED

```

GLOBAL EQUATES

```

0001 =    IDEN    EQU      1        ;IDENTIFIER
0002 =    NUMB    EQU      2        ;NUMBER
0003 =    STRNG   EQU      3        ;STRING
0004 =    SPECL   EQU      4        ;SPECIAL CHARACTER

0001 =    PLABT   EQU      0001B    ;PROGRAM LABEL
0002 =    DLABT   EQU      0010B    ;DATA LABEL
0004 =    EQUT    EQU      0100B    ;EQUATE
0005 =    SETT    EQU      0101B    ;SET
0006 =    MACT    EQU      0110B    ;MACRO

0008 =    EXTT    EQU      1000B    ;EXTERNAL
000B =    REFT    EQU      1011B    ;REFER

```

CP/M v.2.2

COPYRIGHT © 1980

DIGITAL RESEARCH, INC.

P. O. BOX 579
 PACIFIC GROVE, CA 93950

SERIAL # L-756

```

000C =      GLBT   EQU    1100B   ;GLOBAL
;
;
000D =      CR     EQU    0DH
;
;      DATA AREAS
;      SYMBOL TABLE BEGINS AT THE END OF THIS MODULE
0005 =      FIXD   EQU    5       ;5 BYTES OVERHEAD WITH EACH SYMBOL ENTRY
;                                     2BY COLLISION, 1BY TYPE/LEN, 2BY VALUE
0080 =      HSIZE  EQU    128     ;HASH TABLE SIZE
007F =      HMASK  EQU    HSIZE-1 ;HASH MASK FOR CODING
135B      HASHT:  DS      HSIZE*2 ;HASH TABLE
145B      HASHC:  DS      1       ;HASH CODE AFTER CALL ON LOOKUP

```

SYMBOL TABLE ENTRY FORMAT IS

```

-----
: HIGH VAL BYTE :
-----
: LOW VAL BYTE :
-----
: CHARACTER N   :
-----
: ...          :
-----
: CHARACTER 1   :
-----
: TYPE  : LENG :
-----
: HIGH COLLISION:
-----
SYADR= : LOW COLLISION :
-----

```

COPYRIGHT © 1980
 DIGITAL RESEARCH, INC.
 P. O. BOX 579
 PACIFIC GROVE, CA 93950
 SERIAL # _____

WHERE THE LOW/HIGH COLLISION FIELD ADDRESSES ANOTHER ENTRY WITH THE SAME HASH CODE (OR ZERO IF THE END OF CHAIN), TYPE DESCRIBES THE ENTRY TYPE (GIVEN BELOW), LENG IS THE NUMBER OF CHARACTERS IN THE SYMBOL PRINTNAME -1 (I.E., LENG=0 IS A SINGLE CHARACTER PRINT NAME, WHILE LENG=15 INDICATES A 16 CHARACTER NAME). CHARACTER 1 THROUGH N GIVE THE PRINTNAME CHARACTERS IN ASCII UPPER CASE (ALL LOWER CASE NAMES ARE TRANSLATED ON INPUT), AND THE LOW/HIGH VALUE GIVE THE PARTICULAR ADDRESS OR CONSTANT VALUE ASSOCIATED WITH THE NAME. THE REPRESENTATION OF MACROS DIFFERS IN THE FIELDS WHICH FOLLOW THE VALUE FIELD (MACROS ARE NOT CURRENTLY IMPLEMENTED).

THE TYPE FIELD CONSISTS OF FOUR BITS WHICH ARE ASSIGNED AS FOLLOWS:

```

0000      UNDEFINED SYMBOL
0001      LOCAL   LABELLED PROGRAM
0010      LOCAL   LABELLED DATA
0011      (UNUSED)
0100      EQUATE
0101      SET
0110      MACRO
0111      (UNUSED)

```

```

;
;      1000      (UNUSED)
;      1001      EXTERN LABELLED PROGRAM
;      1010      EXTERN LABELLED DATA
;      1011      REFERENCE TO MODULE
;      1100      (UNUSED)
;      1101      GLOBAL UNDEFINED SYMBOL
;      1110      GLOBAL LABELLED PROGRAM
;      1111      (UNUSED)
;

```

TYPE DEFINITIONS

```

0001 = PLABT EQU 0001B ;PROGRAM LABEL
0002 = DLABT EQU 0010B ;DATA LABEL
0004 = EQUT EQU 0100B ;EQUATE
0005 = SETT EQU 0101B ;SET
0006 = MACT EQU 0110B ;MACRO
;
0008 = EXTT EQU 1000B ;EXTERNAL ATTRIBUTE
000B = REFT EQU 1011B ;REFER
000C = GLBT EQU 1100B ;GLOBAL ATTRIBUTE
;

```

```

INISY: ;INITIALIZE THE SYMBOL TABLE
145C 215B13 LXI H,HASHT ;ZERO THE HASH TABLE
145F 0680 MVI B,HSIZE
1461 AF XRA A ;CLEAR ACCUM
INIO:
1462 77 MOV M,A
1463 23 INX H
1464 77 MOV M,A ;CLEAR DOUBLE WORD
1465 23 INX H
1466 05 DCR B
1467 C26214 JNZ INIO
;

```

```

;
;      SET SYMBOL TABLE POINTERS
146A 210000 LXI H,0
146D 22D601 SHLD SYADR
;
1470 C9 RET
;

```

```

CHASH: ;COMPUTE HASH CODE FOR CURRENT ACCUMULATOR
1471 218801 LXI H,ACCLN
1474 46 MOV B,M ;GET ACCUM LENGTH
1475 AF XRA A ;CLEAR ACCUMULATOR
1476 23 CHO: INX H ;MOVE TO FIRST/NEXT CHARACTER POSITION
1477 86 ADD M ;ADD WITH OVERFLOW
1478 05 DCR B
1479 C27614 JNZ CHO
147C E67F ANI HMASK ;MASK BITS FOR MODULO HZISE
147E 325B14 STA HASHC ;FILL HASHC WITH RESULT
1481 C9 RET
;

```

```

SETLN: ;SET THE LENGTH FIELD OF THE CURRENT SYMBOL
1482 47 MOV B,A ;SAVE LENGTH IN B
1483 2AD601 LHLD SYADR
1486 23 INX H

```

COPYRIGHT © 1980
 DIGITAL RESEARCH, INC.
 P. O. BOX 579
 PACIFIC GROVE, CA 93950
 SERIAL # _____

```

1487 23      INX      H
1488 7E      MOV      A,M      ;GET TYPE/LENGTH FIELD
1489 E6F0    ANI      OFOH     ;MASK OUT TYPE FIELD
148B B0      ORA      B        ;MASK IN LENGTH
148C 77      MOV      M,A
148D C9      RET

```

```

;
GETLN:      ;GET THE LENGTH FIELD TO REG-A

```

```

148E 2AD601  LHLD     SYADR
1491 23      INX      H
1492 23      INX      H
1493 7E      MOV      A,M
1494 E60F    ANI      OFH
1496 3C      INR      A        ;LENGTH IS STORED AS VALUE - 1
1497 C9      RET

```

```

;
FOUND:      ;FOUND RETURNS TRUE IF SYADR IS NOT ZERO (TRUE IS NZ FLAG HERE)

```

```

1498 2AD601  LHLD     SYADR
149B 7D      MOV      A,L
149C B4      ORA      H
149D C9      RET

```

```

;
LOOKUP:     ;LOOK FOR SYMBOL IN ACCUMULATOR
            CALL     CHASH     ;COMPUTE HASH CODE
            ;
            NORMALIZE IDENTIFIER TO 16 CHARACTERS

```

```

149E CD7114  CALL     CHASH
14A1 218801  LXI      H,ACCLN
14A4 7E      MOV      A,M
14A5 FE11    CPI      17
14A7 DAAC14  JC       LENOK
14AA 3610    MVI      M,16

```

```

LENOK:      ;
            LOOK FOR SYMBOL THROUGH HASH TABLE

```

```

14AC 215B14  LXI      H,HASHC
14AF 5E      MOV      E,M
14B0 1600    MVI      D,0      ;DOUBLE HASH CODE IN D,E
14B2 215B13  LXI      H,HASHT     ;BASE OF HASH TABLE
14B5 19      DAD      D
14B6 19      DAD      D        ;HASHT(HASHC)
14B7 5E      MOV      E,M     ;LOW ORDER ADDRESS
14B8 23      INX      H
14B9 66      MOV      H,M
14BA 6B      MOV      L,E     ;HEADER TO LIST OF SYMBOLS IS IN H,L
14BB 22D601  LOOK0:  SHLD     SYADR
14BE CD9814  CALL     FOUND
14C1 C8      RZ              ;RETURN IF SYADR BECOMES ZERO

```

```

;
;
OTHERWISE EXAMINE CHARACTER STRING FOR MATCH

```

```

14C2 CD8E14  CALL     GETLN     ;GET LENGTH TO REG-A
14C5 218801  LXI      H,ACCLN
14C8 BE      CMP      M
14C9 C2E114  JNZ      LCOMP

```

```

;
;
LENGTH MATCH, TRY TO MATCH CHARACTERS

```

```

14CC 47      MOV      B,A     ;STRING LENGTH IN B
14CD 23      INX      H        ;HL ADDRESSES ACCUM
14CE EB      XCHG           ;TO D,E

```

COPYRIGHT © 1980
DIGITAL RESEARCH, INC.
P. O. BOX 579
PACIFIC GROVE, CA 93950

SERIAL # _____

```

14CF 2AD601      LHLD    SYADR
14D2 23          INX     H
14D3 23          INX     H
14D4 23          INX     H      ;ADDRESSES CHARACTERS
14D5 1A          LOOK1: LDAX    D      ;NEXT CHARACTER FROM ACCUM
14D6 BE          CMP     M      ;NEXT CHARACTER IN SYMBOL TABLE
14D7 C2E114      JNZ     LCOMP
;               CHARACTER MATCHED, INCREMENT TO NEXT
14DA 13          INX     D
14DB 23          INX     H
14DC 05          DCR     B
14DD C2D514      JNZ     LOOK1
;
;               COMPLETE MATCH AT CURRENT SYMBOL, SYADR IS SET
14E0 C9          RET
;
;               LCOMP: ;NOT FOUND, MOVE SYADR DOWN ONE COLLISION ADDRESS
14E1 2AD601      LHLD    SYADR
14E4 5E          MOV     E,M
14E5 23          INX     H
14E6 56          MOV     D,M      ;COLLISION ADDRESS IN D,E
14E7 EB          XCHG
14E8 C3BB14      JMP     LOOK0
;
;
;               ENTER: ;ENTER SYMBOL IN ACCUMULATOR
;               ENSURE THERE IS ENOUGH SPACE IN THE TABLE
14EB 218801      LXI     H,ACCLN
14EE 5E          MOV     E,M
14EF 1600        MVI     D,0      ;DOUBLE PRECISION ACCLN IN D,E
14F1 2ACB01      LHLD    SYTOP
14F4 22D601      SHLD    SYADR      ;NEXT SYMBOL LOCATION
14F7 19          DAD     D      ;SYTOP+ACCLN
14F8 110500      LXI     D,FXD      ;FIXED DATA/SYMBOL
14FB 19          DAD     D      ;HL HAS NEXT TABLE LOCATION FOR SYMBOL
14FC EB          XCHG      ;NEW SYTOP IN D,E
14FD 2ACD01      LHLD    SYMAX      ;MAXIMUM SYMTOPI VALUE
1500 7B          MOV     A,E
1501 95          SUB     L      ;COMPUTE 16-BIT DIFFERENCE
1502 7A          MOV     A,D
1503 9C          SBB     H
1504 EB          XCHG      ;NEW SYTOP IN H,L
1505 D24115      JNC     OVERER      ;OVERFLOW IN TABLE
;
;               OTHERWISE NO ERROR
1508 22CB01      SHLD    SYTOP      ;SET NEW TABLE TOP
150B 2AD601      LHLD    SYADR      ;SET COLLISION FIELD
150E EB          XCHG      ;CURRENT SYMBOL ADDRESS TO D,E
150F 215B14      LXI     H,HASHC      ;HASH CODE FOR CURRENT SYMBOL TO H,L
1512 4E          MOV     C,M      ;LOW BYTE
1513 0600        MVI     B,0      ;DOUBLE PRECISION VALUE IN B,C
1515 215B13      LXI     H,HASHT      ;BASE OF HASH TABLE
1518 09          DAD     B
1519 09          DAD     B      ;HASHT(HASHC) IN H,L
;               D,E ADDRESSES CURRENT SYMBOL - CHANGE LINKS
151A 4E          MOV     C,M      ;LOW ORDER OLD HEADER

```

COPYRIGHT © 1980

DIGITAL RESEARCH, INC.

P. O. BOX 579

PACIFIC GROVE, CA 93950

SERIAL #

```

151B 23      INX      H
151C 46      MOV      B,M      ;HIGH ORDER OLD HEADER
151D 72      MOV      M,D      ;HIGH ORDER NEW HEADER TO HASH TABLE
151E 2B      DCX      H
151F 73      MOV      M,E      ;LOW ORDER NEW HEADER TO HASH TABLE
1520 EB      XCHG
1521 71      MOV      M,C      ;LOW ORDER OLD HEADER TO COLLISION FIELD
1522 23      INX      H
1523 70      MOV      M,B      ;HIGH ORDER OLD HEADER TO COLLISION FIELD

```

;

;

HASH CHAIN NOW REPAIRED FOR THIS ENTRY, COPY THE PRINTNAME

```

1524 118801  LXI      D,ACCLN
1527 1A      LDAX     D      ;GET SYMBOL LENGTH
1528 FE11    CPI      17     ;LARGER THAN 16 SYMBOLS?
152A DA2F15  JC      ENT1
152D 2E10    MVI      A,16    ;TRUNCATE TO 16 CHARACTERS
;          COPY LENGTH FIELD, FOLLOWED BY PRINTNAME CHARACTERS
152F 47      ENT1:    MOV      B,A      ;COPY LENGTH TO B
1530 3D      DCR      A      ;1-16 CHANGED TO 0-15
1531 23      INX      H      ;FOLLOWING COLLISION FIELD
1532 77      MOV      M,A      ;STORE LENGTH WITH UNDEFINED TYPE (0000)
1533 23      ENT2:    INX      H
1534 13      INX      D
1535 1A      LDAX     D
1536 77      MOV      M,A      ;STORE NEXT CHARACTER OF PRINTNAME
1537 05      DCR      B      ;LENGTH=LENGTH-1
1538 C23315  JNZ      ENT2     ;FOR ANOTHER CHARACTER

```

;

;

PRINTNAME COPIED, ZERO THE VALUE FIELD

```

153B AF      XRA      A      ;ZERO A
153C 23      INX      H      ;LOW ORDER VALUE
153D 77      MOV      M,A
153E 23      INX      H
153F 77      MOV      M,A      ;HIGH ORDER VALUE
1540 C9      RET

```

;

OVERER: ;OVERFLOW IN SYMBOL TABLE

```

1541 214A15  LXI      H,ERRO
1544 CD1202  CALL     PCON
1547 C31E02  JMP      EOR      ;END OF EXECUTION
154A 53594D424FERRO: DB      'SYMBOL TABLE OVERFLOW',CR

```

;

SETTY: ;SET CURRENT SYMBOL TYPE TO VALUE IN REG-A

```

1560 17      RAL
1561 17      RAL
1562 17      RAL
1563 17      RAL
1564 E6F0    ANI      OFOH    ;TYPE MOVED TO HIGH ORDER 4-BITS
1566 47      MOV      B,A      ;SAVE IT IN B
1567 2AD601  LHLD     SYADR    ;BASE OF SYMBOL TO ACCESS
156A 23      INX      H
156B 23      INX      H      ;ADDRESS OF TYPE/LENGTH FIELD
156C 7E      MOV      A,M      ;GET IT AND MASK
156D E60F    ANI      OFH     ;LEAVE LENGTH
156F B0      ORA      B      ;MASK IN TYPE
1570 77      MOV      M,A      ;STORE IT

```

COPYRIGHT © 1980

DIGITAL RESEARCH, INC.

P. O. BOX 579

PACIFIC GROVE, CA 93950

SERIAL # _____

```

1571 C9          RET
;
GETTY: ;RETURN THE TYPE OF THE VALUE IN CURRENT SYMBOL
1572 2AD601     LHL  SYADR
1575 23         INX  H
1576 23         INX  H
1577 7E         MOV  A,M
1578 1F         RAR
1579 1F         RAR
157A 1F         RAR
157B 1F         RAR
157C E60F       ANI  0FH ;TYPE MOVED TO LOW 4-BITS OF REG-A
157E C9         RET

```

```

;
VALADR: ;GET VALUE FIELD ADDRESS FOR CURRENT SYMBOL
157F CD8E14     CALL GETLN ;PRINTNAME LENGTH TO ACCUM
1582 2AD601     LHL  SYADR ;BASE ADDRESS
1585 5F         MOV  E,A
1586 1600       MVI  D,0
1588 19         DAD  D ;BASE(LEN)
1589 23         INX  H
158A 23         INX  H ;FOR COLLISION FIELD
158B 23         INX  H ;FOR TYPE/LEN FIELD
158C C9         RET ;WITH H,L ADDRESSING VALUE FIELD

```

```

;
SETVAL: ;SET THE VALUE FIELD OF THE CURRENT SYMBOL
; VALUE IS SENT IN H,L
158D E5         PUSH H ;SAVE VALUE TO SET
158E CD7F15     CALL VALADR
1591 D1         POP  D ;POP VALUE TO SET, HL HAS ADDRESS TO FILL
1592 73         MOV  M,E
1593 23         INX  H
1594 72         MOV  M,D ;FIELD SET
1595 C9         RET

```

```

;
GETVAL: ;GET THE VALUE FIELD OF THE CURRENT SYMBOL TO H,L
1596 CD7F15     CALL VALADR ;ADDRESS OF VALUE FIELD TO H,L
1599 5E         MOV  E,M
159A 23         INX  H
159B 56         MOV  D,M
159C EB         XCHG
159D C9         RET

```

```

;
15A0 =          ENDMOD EQU ($ AND OFFEOH) + 20H
159E

```

COPYRIGHT © 1980
 DIGITAL RESEARCH, INC.
 P. O. BOX 573
 PACIFIC GROVE, CA 93950
 SERIAL # _____

0188 ACCLN	0189 ACCUM	0040 ACPAX	01D2 ASPC	1476 CHO
1471 CHASH	000D CR	0002 DLABT	15A0 ENDMOD	152F ENT1
1533 ENT2	14EB ENTER	021E EOR	0004 EQUT	154A ERRO
01C9 EVALUE	0008 EXTT	0005 FIXD	1498 FOUND	01D0 FPC
148E GETLN	1572 GETTY	1596 GETVAL	000C GLBT	145B HASHC
135B HASHT	007F HMASK	0080 HSIZE	0001 IDEN	1462 INIO
145C INISY	0200 IOMOD	14E1 LCOMP	14AC LENOK	14BB LOOKO
14D5 LOOK1	149E LOOKUP	0006 MACT	0002 NUMB	1541 OVERER
01CF PASS	0078 PBMAX	0184 PBP	010C PBUFF	0212 PCON
0001 PLABT	000B REFT	1482 SETLN	0005 SETT	1560 SETTY
158D SETVAL	0004 SPECL	0003 STRNG	01D6 SYADR	01D4 SYBAS
01CD SYMAX	01CB SYTOP	0185 TOKEN	157F VALADR	0186 VALUE

COPYRIGHT © 1980

DIGITAL RESEARCH, INC.

P. O. BOX 573

PACIFIC GROVE, CA 93950

SERIAL # _____